      MODULE myQsortMod
! 構造型のクイックソート
! 問題に合わせてうまく流用するための元プログラム．
! メンバーsortObjについて，インデックスが小さいと小さい（昇順）

! 他の言語だとpairとかにあたるものを作りたい．
      implicit none
      integer, parameter, private :: p = 16
      
      type myData
        ! setter,getterは作らないで，メインプログラムから直接アクセス．
        
        ! ソートの基準
        integer(p) :: sortObj
        ! その他のデータ
        integer(p) :: a
      end type myData
      
      interface Qsort
        module procedure :: myQsort
      end interface
      
      contains
      pure recursive function myQsort( a ) result( r )
        implicit none
        type(myData), intent(in) :: a(:)
        type(myData) :: r( size(a) )
        integer(p) :: sample(3), len ,i,j
        
        len = size( a )
        if( len<=1 )then
          r = a
        else
          !select pivot
          sample(1:3) = [a(1)%sortObj, a(len/2)%sortObj, a(len)%sortObj]
          do i = 1,2
            do j = 2,3
              if( sample(i)>sample(j) )then
                len = sample(i) !len is now buffer
                sample(i) = sample(j)
                sample(j) = len
              end if
            end do
          end do
          
          r = [ myQsort( pack(a(:), a(:)%sortObj< sample(2)) ), &
                         pack(a(:), a(:)%sortObj==sample(2))  , &
                myQsort( pack(a(:), a(:)%sortObj> sample(2)) ) ]
        end if
      end function myQsort
      end MODULE myQsortMod
      
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

      PROGRAM Megalomania
      use myQsortMod
      IMPLICIT NONE
      INTEGER,PARAMETER :: prec = 16
      integer(prec) :: n
      type(myData),allocatable :: ab(:)
      integer(prec) :: i,time
      
      read*,n
      allocate( ab(n) )
      do i = 1,n
        read*,ab(i)%a,ab(i)%sortObj
      end do
      
      ab = qsort( ab )
      
      time = 0
      do i = 1,n
        !finish work
        time = time + ab(i)%a
        if( time>ab(i)%sortObj )then
          print*,'No';stop
        end if
        
      end do
      
      print*,'Yes'
      
      
      
      stop
      !debugg
      
      
      
      END PROGRAM